Gessel's walks

$$G(x,y,t) := \sum_{n\geq 0}\sum_{i,j} f_{i,j;n} x^i y^j t^n$$



- 79 inequivalent step sets;
- long history of special cases;
- Gessel's was left;
- conjectured **not** soln LDE.



Thm. [Bostan-Kauers 2010]

G is algebraic! (and thus soln LDE)

Computer-driven discovery and proof

Gessel's walks



$$G(x,y,t) := \sum_{n\geq 0} \sum_{i,j} f_{i,j;n} x^i y^j t^n$$



- compute G up to t^{1000};
- conjecture holonomic (LDE with 1.5 billion coeffs!); Week 5
- check for sanity (bit size, more coeffs, Fuchsian, p-curvature);
- Oho!
- Conjecture polynomials (deg ≤(45,45,25), 25 digit coeffs);
- Proof by (big) resultants. Next week

minimal pol. ≈30 Gb

G is algebraic!

Computer-driven discovery and proof

SIAM Flea



Problem 6. *A flea starts at (0, 0) on the infinite two-dimensional integer lattice and executes a biased random walk: At each step it hops north or south with probability 1/4, east with probability 1/4 + c, and west with probability 1/4 − c. The probability that the flea returns to (0, 0) sometime during its wanderings is 1/2. What is c?*



Two quantities of interest: $\begin{cases} p(c) & := \text{Prob(return to 0)}, \\ q_n(c) & := \text{Prob(at 0 at step } 2n) \,. \end{cases}$

$$\mathbb{E}(\#\text{ returns}) = \sum_{k=1}^{\infty} k p(c)^k (1 - p(c)) = \frac{p(c)}{1 - p(c)} = \sum_{n=1}^{\infty} q_n(c)$$

Key: a binomial sum

$$q_n(c) := \sum_{k=0}^{2n} \binom{2n}{2k} \binom{2k}{k} \binom{2n-2k}{n-k} \left(\frac{1}{4} + c\right)^k \left(\frac{1}{4} - c\right)^k \left(\frac{1}{4}\right)^{2n-2k}$$

Ready for computer algebra

```
> U:=binomial(2*n,2*k)*binomial(2*k,k)*binomial(2*n-2*k,n-k)*p[E]
  ^k*p[W]^k*p[N]^(n-k)*p[S]^(n-k);
```

$$U := \binom{2\,n}{2\,k}\binom{2\,k}{k}\binom{2\,n-2\,k}{n-k} p_E^k p_W^k p_N^{n-k} p_S^{n-k} \qquad \textbf{(1)}$$

This is ``summed" by Zeilberger's algorithm that will be presented later in the course

```
> Z:=SumTools[Hypergeometric][Zeilberger](U,n,k,S[n]);
```

$$Z := \Big[S_n^2 \left(n^2+4\,n+4\right) + \left(-8\,n^2\,p_E p_W - 8\,n^2\,p_N p_S - 24\,n\,p_E p_W - 24\,n\,p_N p_S - 18\,p_E p_W - 18\,p_S p_N\right) S_n + 16\,n^2\,p_E^2 p_W^2 - 32\,n^2\,p_E p_N p_S p_W + 16\,n^2\,p_N^2 p_S^2 + 32\,n\,p_E^2 p_W^2 - 64\,n\,p_E p_N p_S p_W + 32\,n\,p_N^2 p_S^2 + 12\,p_E^2 p_W^2 - 24\,p_E p_N p_S p_W + 12\,p_S^2 p_N^2,$$

$$\frac{1}{(-n+k-1)^2\,(-n+k-2)^2}\Bigg(\Bigg(\frac{1}{n^2+4\,n+4}\Big(4\,k^2\left(4\,n^2\,p_E p_W - 4\,n^2\,p_N p_S + 8\,n\,p_E p_W - 8\,n\,p_N p_S + 3\,p_E p_W - 3\,p_S p_N\right)\Big) - \frac{8\,k\left(4\,n^2+8\,n+3\right)\left(n\,p_E p_W - 2\,n\,p_N p_S + 2\,p_E p_W - 3\,p_S p_N\right)}{n^2+4\,n+4} + \frac{1}{n+2}\Big(4\left(4\,n^3\,p_E p_W - 16\,n^3\,p_N p_S + 16\,n^2\,p_E p_W - 52\,n^2\,p_N p_S + 19\,n\,p_E p_W - 52\,n\,p_N p_S + 6\,p_E p_W - 15\,p_S p_N\right)\Big)\Bigg) k^2\,p_N p_S \binom{2\,n}{2\,k}\binom{2\,k}{k}\binom{2\,n-2\,k}{n-k} p_E^k p_W^k p_N^{n-k} p_S^{n-k}\left(n^2+4\,n+4\right)\Bigg)\Big] \qquad \textbf{(2)}$$

Explanation of the output:

```
> Z[1];
```

$$S_n^2\left(n^2+4\,n+4\right) + \left(-8\,n^2\,p_E p_W - 8\,n^2\,p_N p_S - 24\,n\,p_E p_W - 24\,n\,p_N p_S - 18\,p_E p_W - 18\,p_S p_N\right) S_n + 16\,n^2\,p_E^2 p_W^2 - 32\,n^2\,p_E p_N p_S p_W + 16\,n^2\,p_N^2 p_S^2 + 32\,n\,p_E^2 p_W^2 - 64\,n\,p_E p_N p_S p_W + 32\,n\,p_N^2 p_S^2 + 12\,p_E^2 p_W^2 - 24\,p_E p_N p_S p_W + 12\,p_S^2 p_N^2 \qquad \textbf{(3)}$$

is a shift operator corresponding to the linear recurrence

```
> rec:=add(coeff(%,S[n],i)*q(n+i),i=0..degree(%,S[n]));
```

$$rec := \left(16\,n^2\,p_E^2 p_W^2 - 32\,n^2\,p_E p_N p_S p_W + 16\,n^2\,p_N^2 p_S^2 + 32\,n\,p_E^2 p_W^2 - 64\,n\,p_E p_N p_S p_W + 32\,n\,p_N^2 p_S^2 + 12\,p_E^2 p_W^2 - 24\,p_E p_N p_S p_W + 12\,p_S^2 p_N^2\right) q(n) + \left(-8\,n^2\,p_E p_W - 8\,n^2\,p_N p_S - 24\,n\,p_E p_W - 24\,n\,p_N p_S - 18\,p_E p_W - 18\,p_S p_N\right) q(n+1) + \left(n^2+4\,n+4\right) q(n+2) \qquad \textbf{(4)}$$

The output of the algorithm is a telescoping identity concerning the sequence U(n,k):

```
> eval(rec,q=unapply(U,n))-(subs(k=k+1,Z[2])-Z[2]);
```

$$\begin{aligned}
&\Big(16 n^2 p_E^2 p_W^2 - 32 n^2 p_E p_N p_S p_W + 16 n^2 p_N^2 p_S^2 + 32 n p_E^2 p_W^2 - 64 n p_E p_N p_S p_W + 32 n p_N^2 p_S^2 + 12 p_E^2 p_W^2 - 24 p_E p_N p_S p_W + 12 p_S^2 p_N^2\Big) \binom{2n}{2k} \binom{2k}{k} \binom{2n-2k}{n-k} p_E^k p_W^k p_N^{n-k} p_S^{n-k} \\
&+ \Big(-8 n^2 p_E p_W - 8 n^2 p_N p_S - 24 n p_E p_W - 24 n p_N p_S - 18 p_E p_W - 18 p_S p_N\Big) \binom{2n+2}{2k} \binom{2k}{k} \binom{2n-2k+2}{n+1-k} p_E^k p_W^k p_N^{n+1-k} p_S^{n+1-k} \\
&+ \big(n^2 + 4n + 4\big) \binom{2n+4}{2k} \binom{2k}{k} \binom{2n+4-2k}{n+2-k} p_E^k p_W^k p_N^{n+2-k} p_S^{n+2-k} \\
&- \frac{1}{(-n+k)^2 (-n+k-1)^2} \Bigg(\Bigg(\frac{1}{n^2+4n+4} \Big(4 (k+1)^2 \big(4 n^2 p_E p_W - 4 n^2 p_N p_S + 8 n p_E p_W - 8 n p_N p_S + 3 p_E p_W - 3 p_S p_N\big)\Big) \\
&\quad - \frac{8 (k+1) \big(4 n^2 + 8 n + 3\big) \big(n p_E p_W - 2 n p_N p_S + 2 p_E p_W - 3 p_S p_N\big)}{n^2+4n+4} \\
&\quad + \frac{1}{n+2} \Big(4 \big(4 n^3 p_E p_W - 16 n^3 p_N p_S + 16 n^2 p_E p_W - 52 n^2 p_N p_S + 19 n p_E p_W - 52 n p_N p_S + 6 p_E p_W - 15 p_S p_N\big)\Big) \Bigg) \\
&\quad (k+1)^2 p_N p_S \binom{2n}{2k+2} \binom{2k+2}{k+1} \binom{2n-2k-2}{n-k-1} p_E^{k+1} p_W^{k+1} p_N^{n-k-1} p_S^{n-k-1} \big(n^2+4n+4\big) \Bigg) \\
&+ \frac{1}{(-n+k-1)^2 (-n+k-2)^2} \Bigg(\Bigg(\frac{1}{n^2+4n+4} \Big(4 k^2 \big(4 n^2 p_E p_W - 4 n^2 p_N p_S + 8 n p_E p_W - 8 n p_N p_S + 3 p_E p_W - 3 p_S p_N\big)\Big) \\
&\quad - \frac{8 k \big(4 n^2 + 8 n + 3\big) \big(n p_E p_W - 2 n p_N p_S + 2 p_E p_W - 3 p_S p_N\big)}{n^2+4n+4} \\
&\quad + \frac{1}{n+2} \Big(4 \big(4 n^3 p_E p_W - 16 n^3 p_N p_S + 16 n^2 p_E p_W - 52 n^2 p_N p_S + 19 n p_E p_W - 52 n p_N p_S + 6 p_E p_W - 15 p_S p_N\big)\Big) \Bigg) \\
&\quad k^2 p_N p_S \binom{2n}{2k} \binom{2k}{k} \binom{2n-2k}{n-k} p_E^k p_W^k p_N^{n-k} p_S^{n-k} \big(n^2+4n+4\big) \Bigg)
\end{aligned} \tag{5}$$

```
> normal(expand(%));
```

$$0 \tag{6}$$

thus rec is a recurrence satisfied by the sequence q(n). (It does not have a nice solution.)

Initial conditions:

```
> iniq:={q(0)=1,q(1)=2*p[N]*p[S]+2*p[E]*p[W]};
```

$$iniq := \{q(0) = 1, q(1) = 2\, p_E p_W + 2\, p_S p_N\} \quad (7)$$

Now, we want to sum q_n(c). In other terms, we want the value at z=1 of

```
> Sum(q(n)*z^n,n=0..infinity);
```

$$\sum_{n=0}^{\infty} q(n)\, z^n \quad (8)$$

This power series satisfies a linear differential equation

```
> gfun:-rectodiffeq({rec} union iniq,q(n),y(z));
```

$$\Big\{\left(12\, z\, p_E^2 p_W^2 - 24\, z\, p_E p_N p_S p_W + 12\, z\, p_N^2 p_S^2 - 2\, p_E p_W - 2\, p_S p_N\right) y(z) + \left(48\, z^2\, p_E^2 p_W^2 - 96\, z^2\, p_E p_N p_S p_W + 48\, z^2\, p_N^2 p_S^2 - 16\, z\, p_E p_W - 16\, z\, p_N p_S + 1\right) \left(\frac{d}{dz}\, y(z)\right) + \left(16\, z^3\, p_E^2 p_W^2 - 32\, z^3\, p_E p_N p_S p_W + 16\, z^3\, p_N^2 p_S^2 - 8\, z^2\, p_E p_W - 8\, z^2\, p_N p_S + z\right) \left(\frac{d^2}{dz^2}\, y(z)\right), y(0) = 1\Big\} \quad (9)$$

That one has a classical solution:

```
> dsolve(%,y(z)) assuming p[E]>0,p[E]<1,p[W]>0,p[W]<1,p[N]>0,p[N]
  <1,p[S]>0,p[S]<1;
```

$$y(z) = \Bigg(\Bigg(-\_C2 - \frac{2\, I \sqrt{2}}{2 \sqrt{p_E} \sqrt{p_N} \sqrt{p_S} \sqrt{p_W} + p_E p_W + p_N p_S}\Bigg) \sqrt{8\, z\, p_E p_W + 8\, z\, p_N p_S + 16\, z \sqrt{p_E} \sqrt{p_N} \sqrt{p_S} \sqrt{p_W} - 2 \Bigg(-\frac{\sqrt{p_E} \sqrt{p_N} \sqrt{p_S} \sqrt{p_W}}{2} + (p_E p_W - p_N p_S)^2\, z - \frac{p_E p_W}{4} - \frac{p_N p_S}{4}\Bigg)}$$
$$\mathrm{hypergeom}\Bigg(\left[\frac{1}{2}, \frac{1}{2}\right], [1], -\left(16\, (p_E p_W - p_N p_S)^2 \sqrt{p_E} \sqrt{p_N} \sqrt{p_S} \sqrt{p_W}\, z\right) \Big/ \left(\left(p_E p_W + p_N p_S - 2 \sqrt{p_E} \sqrt{p_N} \sqrt{p_S} \sqrt{p_W}\right) \left(4\, z\, p_E^2 p_W^2 - 8\, z\, p_E p_N p_S p_W + 4\, z\, p_N^2 p_S^2 - p_E p_W - p_N p_S - 2 \sqrt{p_E} \sqrt{p_N} \sqrt{p_S} \sqrt{p_W}\right)\right)\Bigg) \Bigg/$$
$$\Bigg(\sqrt{1 + 16\, (p_E p_W - p_N p_S)^2\, z^2 + (-8\, p_E p_W - 8\, p_N p_S)\, z}\, \Big(-8\, z \sqrt{p_E} \sqrt{p_N} \sqrt{p_S} \sqrt{p_W} - 1 + (4\, p_E p_W + 4\, p_N p_S)\, z\Big)\Bigg)$$

$$+ \left(\_C2 \sqrt{8 z p_E p_W + 8 z p_N p_S + 16 z \sqrt{p_E}\sqrt{p_N}\sqrt{p_S}\sqrt{p_W} - 2} \left(-\frac{\sqrt{p_E}\sqrt{p_N}\sqrt{p_S}\sqrt{p_W}}{2} + (p_E\right.\right.$$

$$\left. - p_N p_S)^2 \sqrt{p_E}\sqrt{p_N}\sqrt{p_S}\sqrt{p_W}\, z\right) \Big/ \left(\left(p_E p_W + p_N p_S\right.\right.$$

$$- 2\sqrt{p_E}\sqrt{p_N}\sqrt{p_S}\sqrt{p_W}\Big) \left(4 z p_E^2 p_W^2 - 8 z p_E p_N p_S p_W + 4 z p_N^2 p_S^2 - p_E p_W\right.$$

$$\left.\left.\left. - p_N p_S - 2\sqrt{p_E}\sqrt{p_N}\sqrt{p_S}\sqrt{p_W}\right)\right)\right) \Big/$$

$$\left(\sqrt{1 + 16 (p_E p_W - p_N p_S)^2 z^2 + (-8 p_E p_W - 8 p_N p_S) z} \right. \Big($$

$$\left. -8 z \sqrt{p_E}\sqrt{p_N}\sqrt{p_S}\sqrt{p_W} - 1 + (4 p_E p_W + 4 p_N p_S) z\Big)\right)$$

```
> sol:=op(2,%):
> sol:=simplify(sol);
```

$$sol := \left(4\,\mathrm{I} \sqrt{8 z p_E p_W + 8 z p_N p_S + 16 z \sqrt{p_E}\sqrt{p_N}\sqrt{p_S}\sqrt{p_W} - 2}\, \sqrt{2} \right.\Bigg($$

$$-\frac{\sqrt{p_E}\sqrt{p_N}\sqrt{p_S}\sqrt{p_W}}{2} + (p_E p_W - p_N p_S)^2 z - \frac{p_E p_W}{4} - \frac{p_N p_S}{4}\Bigg)$$

$$\mathrm{EllipticK}\left(\left(4 (p_E p_W - p_N p_S) p_E^{1/4} p_N^{1/4} p_S^{1/4} p_W^{1/4} \sqrt{z}\right) \Big/\right.$$

$$\left(\sqrt{-\left(\sqrt{p_E}\sqrt{p_W} - \sqrt{p_N}\sqrt{p_S}\right)^2} \left(-4 \left(\sqrt{p_E}\sqrt{p_W} + \sqrt{p_N}\sqrt{p_S}\right)^2 \left(-2 z \sqrt{p_E}\sqrt{p_N}\sqrt{p_S}\sqrt{p_W} - \right.\right.\right.$$

$$\left.\left.\left. + (p_E p_W + p_N p_S) z\right)\right)\right.$$

$$\left.\left.^{1/2} \sqrt{\frac{8 z \sqrt{p_E}\sqrt{p_N}\sqrt{p_S}\sqrt{p_W} - 1 + (4 p_E p_W + 4 p_N p_S) z}{-8 z \sqrt{p_E}\sqrt{p_N}\sqrt{p_S}\sqrt{p_W} - 1 + (4 p_E p_W + 4 p_N p_S) z}}\right)\right)\right) \Big/$$

$$\left(\sqrt{\frac{8 z \sqrt{p_E}\sqrt{p_N}\sqrt{p_S}\sqrt{p_W} - 1 + (4 p_E p_W + 4 p_N p_S) z}{-8 z \sqrt{p_E}\sqrt{p_N}\sqrt{p_S}\sqrt{p_W} - 1 + (4 p_E p_W + 4 p_N p_S) z}}\right.$$

$$\sqrt{1 + 16 (p_E p_W - p_N p_S)^2 z^2 + (-8 p_E p_W - 8 p_N p_S) z}\, \Big(2\sqrt{p_E}\sqrt{p_N}\sqrt{p_S}\sqrt{p_W}$$

$$\left. + p_E p_W + p_N p_S\Big) \pi \left(8 z \sqrt{p_E}\sqrt{p_N}\sqrt{p_S}\sqrt{p_W} - 4 z p_E p_W - 4 z p_N p_S + 1\right)\right)$$

Value at 1:

```
> Final:=eval(sol,z=1);
```

$$Final := \Bigg(4\,I\,\sqrt{8\,p_E p_W + 8\,p_N p_S + 16\,\sqrt{p_E}\,\sqrt{p_N}\,\sqrt{p_S}\,\sqrt{p_W} - 2\,\sqrt{2}\,\Big(-\frac{\sqrt{p_E}\,\sqrt{p_N}\,\sqrt{p_S}\,\sqrt{p_W}}{2} + (p_E p_W - p_N p_S)^2 - \frac{p_E p_W}{4} - \frac{p_N p_S}{4}\Big)} \tag{12}$$

$$\mathrm{EllipticK}\Bigg(\Big(4\,(p_E p_W - p_N p_S)\,p_E^{1/4}\,p_N^{1/4}\,p_S^{1/4}\,p_W^{1/4}\Big) \Bigg/ \Bigg(\sqrt{-\big(\sqrt{p_E}\,\sqrt{p_W} - \sqrt{p_N}\,\sqrt{p_S}\big)^2}$$

$$\Big(-4\,\big(\sqrt{p_E}\,\sqrt{p_W} + \sqrt{p_N}\,\sqrt{p_S}\big)^2\,\Big(-2\,\sqrt{p_E}\,\sqrt{p_N}\,\sqrt{p_S}\,\sqrt{p_W} - \frac{1}{4} + p_E p_W$$

$$+ p_N p_S\Big)\Big)^{1/2}\,\sqrt{\frac{8\,\sqrt{p_E}\,\sqrt{p_N}\,\sqrt{p_S}\,\sqrt{p_W} - 1 + 4\,p_E p_W + 4\,p_N p_S}{-8\,\sqrt{p_E}\,\sqrt{p_N}\,\sqrt{p_S}\,\sqrt{p_W} - 1 + 4\,p_E p_W + 4\,p_N p_S}}\Bigg)\Bigg)\Bigg) \Bigg/$$

$$\Bigg(\sqrt{\frac{8\,\sqrt{p_E}\,\sqrt{p_N}\,\sqrt{p_S}\,\sqrt{p_W} - 1 + 4\,p_E p_W + 4\,p_N p_S}{-8\,\sqrt{p_E}\,\sqrt{p_N}\,\sqrt{p_S}\,\sqrt{p_W} - 1 + 4\,p_E p_W + 4\,p_N p_S}}$$

$$\sqrt{1 + 16\,(p_E p_W - p_N p_S)^2 - 8\,p_E p_W - 8\,p_N p_S}\,\Big(2\,\sqrt{p_E}\,\sqrt{p_N}\,\sqrt{p_S}\,\sqrt{p_W} + p_E p_W$$

$$+ p_N p_S\Big)\,\pi\,\Big(8\,\sqrt{p_E}\,\sqrt{p_N}\,\sqrt{p_S}\,\sqrt{p_W} - 4\,p_E p_W - 4\,p_N p_S + 1\Big)\Bigg)$$

Insert our probabilities:

```
> simplify(eval(Final,[p[W]=1/4+c,p[E]=1/4-c,p[N]=1/4,p[S]=1/4]))
  assuming c>=0,c<=1/4;
```

$$\Bigg(\sqrt{8\,c^2 + 1 - \sqrt{1-4\,c}\,\sqrt{4\,c+1}}\,\Big(-32\,c^4 - 8\,c^2 + \sqrt{1-4\,c}\,\sqrt{4\,c+1} \tag{13}$$

$$+ 1\Big)$$

$$\mathrm{EllipticK}\Bigg(\frac{I\,(1-4\,c)^{1/4}\,\sqrt{2}\,(4\,c+1)^{1/4}\,\sqrt{\sqrt{1-4\,c}\,\sqrt{4\,c+1} + 8\,c^2 + 1}}{\sqrt{8\,c^2 - \sqrt{-16\,c^2+1} + 1}\,\sqrt{\sqrt{-16\,c^2+1} + 8\,c^2 + 1}}\Bigg)\Bigg)$$

$$\Big/ \left(\sqrt{8c^2 - \sqrt{-16c^2+1} + 1}\,\sqrt{\sqrt{-16c^2+1} + 8c^2 + 1}\,c\sqrt{2c^2+1}\,\left(\sqrt{1-4c}\,\sqrt{4c+1} - 8c^2 + 1\right)\pi\right)$$

Simplify further:

```
> Result:=convert(%,hypergeom);
```

$$Result := \left(\sqrt{8c^2 + 1 - \sqrt{1-4c}\,\sqrt{4c+1}}\,\left(-32c^4 - 8c^2 + \sqrt{1-4c}\,\sqrt{4c+1} + 1\right)\mathrm{hypergeom}\left(\left[\frac{1}{2}, \frac{1}{2}\right], [1], \frac{\left(-8c^2-1\right)\sqrt{4c+1}\,\sqrt{1-4c} + 16c^2 - 1}{32c^4 + 16c^2}\right)\right) \Big/ \left(2\sqrt{8c^2 - \sqrt{-16c^2+1} + 1}\,\sqrt{\sqrt{-16c^2+1} + 8c^2 + 1}\,c\sqrt{2c^2+1}\,\left(\sqrt{1-4c}\,\sqrt{4c+1} - 8c^2 + 1\right)\right) \quad \textbf{(14)}$$

```
> plot(Result,c=0..1/4);
```




```
> Digits:=100:
> fsolve(Result=2,c,0..1/4);
```

0.0619139544739909428481752164732121769996387749983620760614672588599310102\
9759615845907105645752087861 **(15)**

This is the desired value, with 100 digits of accuracy.

I. Definition of the Hermite polynomials

```
> R[1] := {H[x](0) = 1, H[x](1) = 2*x, H[x](n+2) = (-2*n-2)*H[x](n)+2*H[x](n+1)*x};
```

$$R_1 := \{H_x(0) = 1, H_x(1) = 2x, H_x(n+2) = (-2n-2)\,H_x(n) + 2\,H_x(n+1)\,x\} \qquad \textbf{(1.1)}$$

```
> R[2] := subs(x = y, R[1]);
```

$$R_2 := \{H_y(0) = 1, H_y(1) = 2y, H_y(n+2) = (-2n-2)\,H_y(n) + 2\,H_y(n+1)\,y\} \qquad \textbf{(1.2)}$$

II. Product

```
> R[3] := gfun:-poltorec(H[x](n)*H[y](n)*v(n), [R[1], R[2], {v(n+1)*(n+1) = v(n), v(1) = 1}], [H[x](n), H[y](n), v(n)], c(n));
```

$$R_3 := \Big\{(16n+16)\,c(n) - 16\,x\,y\,c(n+1) + \left(8x^2 + 8y^2 - 8n - 20\right)c(n+2) - 4\,c(n+3)\,x\,y + (n+4)\,c(n+4), c(0) = 1, c(1) = 4xy, c(2) = 8x^2y^2 - 4x^2 - 4y^2 + 2, c(3) = \frac{32}{3}x^3y^3 - 16x^3y - 16xy^3 + 24xy, c(4) = \frac{32}{3}x^4y^4 - 32x^4y^2 - 32x^2y^4 + 8x^4 + 96x^2y^2 + 8y^4 - 24x^2 - 24y^2 + 6\Big\} \qquad \textbf{(2.1)}$$

III. Differential Equation

```
> gfun:-rectodiffeq(R[3], c(n), f(u));
```

$$\left\{\left(-16u^2xy + 16u^3 + 8ux^2 + 8uy^2 - 4xy - 4u\right)f(u) + \left(16u^4 - 8u^2 + 1\right)\left(\frac{d}{du}f(u)\right), f(0) = 1\right\} \qquad \textbf{(3.1)}$$

```
> dsolve(%,f(u)) assuming 0<u,u<1/2;
```

$$f(u) = \frac{e^{-\frac{4xyu - x^2 - y^2}{(2u-1)(2u+1)}}\sqrt{\frac{1}{(-2u+1)(2u+1)}}}{e^{-x^2 - y^2}} \qquad \textbf{(3.2)}$$